

Eryk Wdowiak  
original: 29 April 2014

Quantitative Analysis  
estimating parameters of Cauchy distribution

First, load the "distrib" package.

```
(%i1) load(distrib)$
```

Now, randomly draw 50 values from the standard Cauchy.

```
(%i2) N:50$  
x:random_cauchy(0,1,N)$
```

Set up the log-likelihood function.

```
(%i4) loglik(mu,gamma):= -N*log(%pi*gamma) -  
sum( log(1+(((x[i]-mu)/gamma)^2)),i,1,N);
```

```
(%o4) loglik( $\mu$ ,  $\Gamma$ ) := (-N) log( $\pi \Gamma$ ) -  $\sum_{i=1}^N \log\left(1 + \left(\frac{x_i - \mu}{\Gamma}\right)^2\right)$ 
```


Maximize it with respect to mu and gamma.

```
(%i5) sol:lbfgs(-loglik(mu,gamma),'[mu,gamma],[0.01,0.99],0.0001,[-1,0])$  
mu_max:subst(sol[1],mu)$  
gamma_max:subst(sol[2],gamma)$  
  
print("")$  
print(mu," = ",mu_max)$  
print(gamma," = ",gamma_max)$  
print("")$
```

```
 $\mu$  = 0.080902253273153  
 $\Gamma$  = 1.036534260262574
```

To see the maximum, plot the log likelihood function in "mu-gamma" space.

```
(%i12) wxplot3d(loglik(mu,gamma),[mu,-0.5,0.5],[gamma,0.5,2.5])$
(%t12)
```



Check to see if second-order conditions are satisfied.

```
(%i13) /* set up the Hessian matrix */
dxx(mu,gamma):=''(diff(diff(loglik(mu,gamma),mu),mu))$
dyy(mu,gamma):=''(diff(diff(loglik(mu,gamma),gamma),gamma))$
dxy(mu,gamma):=''(diff(diff(loglik(mu,gamma),mu),gamma))$
H:matrix(
    [dxx(mu_max,gamma_max),dxy(mu_max,gamma_max)],
    [dxy(mu_max,gamma_max),dyy(mu_max,gamma_max)])$
```

```
(%i17) print("")$
      print("own-partial must be negative:")$
      print("")$
      print("d^2 loglik(mu,gamma)"/"(d mu)^2", " = ", dxx(mu_max,gamma_max))$
      print("")$
      print("d^2 loglik(mu,gamma)"/"(d gamma)^2", " = ", dyy(mu_max,gamma_max))$
      print("")$
```

*own-partial must be negative:*

$$\frac{d^2 \loglik(mu,gamma)}{(d mu)^2} = -22.2581034728271$$

$$\frac{d^2 \loglik(mu,gamma)}{(d gamma)^2} = -24.27935715351495$$

```
(%i24) print("")$
      print("the cross-partial:")$
      print("")$
      print("d^2 loglik(mu,gamma)"/"d mu d gamma", " = ", dxy(mu_max,gamma_max))$
      print("")$
```

*the cross-partial:*

$$\frac{d^2 \loglik(mu,gamma)}{d mu d gamma} = 2.664077718337032$$

```
(%i29) print("")$
      print("the Hessian matrix:")$
      print("H = ",H)$
      print("")$
      print("determinant of Hessian must be positive")$
      print("det(H) = ",determinant(H))$
      print("")$
```

*the Hessian matrix:*

$$H = \begin{bmatrix} -22.2581034728271 & 2.664077718337032 \\ 2.664077718337032 & -24.27935715351495 \end{bmatrix}$$

*determinant of Hessian must be positive*

$$\det(H) = 533.3151336873207$$

```
(%i36) info:-1*invert(H)$
```

```
print("")$  
print("the information matrix:")$  
print("-1*(H^-1) = ",info)$  
print("")$  
print(mu," : ",mu_max,"    se:",sqrt(info[1,1]))$  
print(gamma," : ",gamma_max,"    var:",info[2,2])$  
print("")$
```

*the information matrix:*

$$-1*(H^{-1}) = \begin{bmatrix} 0.045525348185131 & 0.0049953161837312 \\ 0.0049953161837312 & 0.04173536820329 \end{bmatrix}$$

$\mu$  : 0.080902253273153      se: 0.21336669886637  
 $\Gamma$  : 1.036534260262574      var: 0.04173536820329